
Appendix A: Batch FTP


The following example files follow:


CL command definition allowing specification of


the location of a stream of input FTP commands


the location to which the output from the FTP operation is to be written


the TCP/IP address of the destination system


CL command processing program


Example stream of FTP commands


Example output from the batch FTP operation


CL command definition


             CMD        PROMPT('FTP Batch Processing')


             PARM       KWD(INPUTFILE) TYPE(QUAL1) MIN(1) +


                          PROMPT('File containing FTP commands:')


             PARM       KWD(OUTPUTFILE) TYPE(QUAL1) MIN(1) +


                          PROMPT('File to store FTP output:')


             PARM       KWD(HOST) TYPE(*CHAR) LEN(64) MIN(1) +


                          PROMPT('TCP/IP address of destination')


 QUAL1:      QUAL       TYPE(*NAME)


             QUAL       TYPE(*NAME) DFT(*LIBL) SPCVAL((*LIBL)) +


                          PROMPT('Library')


             QUAL       TYPE(*NAME) DFT(*FIRST) SPCVAL((*FIRST)) +


                          PROMPT('Member name')


�
CL program


Note that the FTP command will not normally return an exception if errors are encountered while the FTP command stream is being processed; it is recommended that the FTP output be reviewed (manually or automatically) to check for failures.


/*********************************************************************/


/* Program name:      FTPBATCH                                       */


/* Author:            Mandy Shaw.                                    */


/* Date completed:    15/3/96                                        */


/* Function:          Batch FTP processing.                          */


/* Processing:        Simple program. See code.                      */


/* Input:             File of commands, file for output, host name   */


/*                    to connect to.                                 */


/* Output normal:     None.                                          */


/* Output error:      Return exceptions to caller.                   */


/* Restrictions:      None.                                          */


/*********************************************************************/


PGM (&INPUTFILE &OUTPUTFILE &HOST)


/*********************************************************************/


/* Program parameters. */


DCL &INPUTFILE *CHAR 30


DCL &OUTPUTFILE *CHAR 30


DCL &HOST *CHAR 64


/* Local variables. */


DCL &INPFILOBJ *CHAR 10


DCL &INPFILLIB *CHAR 10


DCL &INPFILMBR *CHAR 10


DCL &OUTFILOBJ *CHAR 10


DCL &OUTFILLIB *CHAR 10


DCL &OUTFILMBR *CHAR 10


/* Message handling. */


DCL &LMSGID *CHAR 7


DCL &LMSGDTA *CHAR 256


DCL &LMSGF *CHAR 10


DCL &LMSGFLIB *CHAR 10


/*********************************************************************/


MONMSG (CPF0001 CPF0006) EXEC(GOTO VALID)


MONMSG (MCH0000 CPF0000) EXEC(GOTO ERROR)


/*********************************************************************/


/* CHGCMD QSYS/FTP ALLOW(*ALL) assumed done */


CHGVAR &INPFILOBJ %SST(&INPUTFILE 1 10)


CHGVAR &INPFILLIB %SST(&INPUTFILE 11 10)


CHGVAR &INPFILMBR %SST(&INPUTFILE 21 10)


CHGVAR &OUTFILOBJ %SST(&OUTPUTFILE 1 10)


CHGVAR &OUTFILLIB %SST(&OUTPUTFILE 11 10)


CHGVAR &OUTFILMBR %SST(&OUTPUTFILE 21 10)


OVRDBF FILE(INPUT) TOFILE(&INPFILLIB/&INPFILOBJ) +


         MBR(&INPFILMBR)


OVRDBF FILE(OUTPUT) TOFILE(&OUTFILLIB/&OUTFILOBJ) +


         MBR(&OUTFILMBR)


FTP RMTSYS(&HOST)


OVRPRTF FILE(QSYSPRT) SPLFNAME(FTPBATCH)


CPYF FROMFILE(&OUTFILLIB/&OUTFILOBJ) +


       TOFILE(QSYSPRT) FROMMBR(&OUTFILMBR)


DLTOVR FILE(*ALL)


RETURN


/*********************************************************************/


VALID: RCVMSG MSGTYPE(*DIAG) MSGDTA(&LMSGDTA) MSGID(&LMSGID) +


              MSGF(&LMSGF) SNDMSGFLIB(&LMSGFLIB)


       SNDPGMMSG MSGID(&LMSGID) MSGF(&LMSGFLIB/&LMSGF) +


                 MSGDTA(&LMSGDTA) MSGTYPE(*DIAG)


ERROR: RCVMSG MSGTYPE(*EXCP) MSGDTA(&LMSGDTA) MSGID(&LMSGID) +


              MSGF(&LMSGF) SNDMSGFLIB(&LMSGFLIB)


          MONMSG CPF0000 EXEC(GOTO PGMERR)


       SNDPGMMSG MSGID(&LMSGID) MSGF(&LMSGFLIB/&LMSGF) +


                 MSGDTA(&LMSGDTA) MSGTYPE(*ESCAPE)


          MONMSG CPF0000 EXEC(GOTO PGMERR)


/*********************************************************************/


PGMERR: SNDPGMMSG MSGID(CPF0001) MSGF(QCPFMSG) +


                    MSGDTA(FTPBATCH) +


                    MSGTYPE(*ESCAPE)


           MONMSG CPF0000


ENDPGM


�
Example FTP command stream


Note that the commands given are merely examples - there are many different FTP commands available, and the format in which, for example, the userid and password is given will differ amongst different destination platforms. This example relates to source and destination systems that are both AS/400s. I have appended a brief explanation to each command used.


<userid> <password>                      (on destination system)


BINARY                                   (specifies data to be sent untranslated)


NAMEFMT 1                                (specifies that data from outside normal


                                         library/object structure is being handled -


                                         in this case, data is to be placed in an AS/400 folder)


CD /QDLS/APPFLR/TEMP                     (specifies destination ‘directory’: an AS/400 folder)


LCD /QSYS.LIB/SHAWM.LIB                  (specifies source ‘directory’: an AS/400 library)


PUT EXAMPLEFIL.FILE EXAMPLE.FIL          (copies file EXAMPLEFIL from library SHAWM on


                                         the source system to document EXAMPLE.FIL in


                                         folder APPFLR/TEMP on the destination


                                         system)


QUIT                                     (indicates FTP processing complete)


�
Example output


Output redirected to a file.


Input read from specified override file.


Connecting to host name PACIFIC3.PSIPLC.PACIFIC.CO.UK at address 170.1.1.1


 using port 21.


220-QTCP at PACIFIC3.PSIPLC.PACIFIC.CO.UK.


220 Connection will close if idle more than 5 minutes.


215  OS/400 is the remote operating system. The TCP/IP version is "V3R1M0".


Enter login ID (shawm):


331 Enter password.


230 SHAWM logged on.


250  Now using naming format "0".


257 "SHAWM" is current library.


Enter an FTP subcommand.


> BINARY


200 Representation type is binary IMAGE.


Enter an FTP subcommand.


> NAMEFMT 1


250  Now using naming format "1".


Server NAMEFMT is 1.


Client NAMEFMT is 1.


Enter an FTP subcommand.


> CD /QDLS/SHAWM/TEMP


250 Current directory changed to /QDLS/SHAWM/TEMP.


Enter an FTP subcommand.


> LCD /QSYS.LIB/SHAWM.LIB


Local working directory is /QSYS.LIB/SHAWM.LIB.


Enter an FTP subcommand.


> PUT ASTOMAC.FILE


250  Now using naming format "1".


257  Current directory is "/QDLS/SHAWM/TEMP".


200 PORT subcommand request successful.


150 Sending file to /QDLS/SHAWM/TEMP/ASTOMAC


250 File transfer completed successfully.


596640 bytes transferred in 10.188 seconds. Transfer rate 58.566 KB/sec.


Enter an FTP subcommand.


> QUIT


221 QUIT subcommand received.

